
Subject: Re: Uttar Pradesh districtwise Neonatal, infant, postnatal and under five Mortality data

Posted by [Bridgette-DHS](#) on Thu, 08 Jun 2023 11:43:29 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

Your request goes far beyond the scope of the forum. Also what you propose to do would not be useful, because of the statistical uncertainty of district-level estimates.

However, I have prepared something that may be useful. I wrote a Stata program to calculate, within the districts of Uttar Pradesh, the proportion of children born in the past five years who had died by the date of the survey, along with 95% confidence intervals adjusted for the survey design. This program could be of interest to other users so I will paste it below (as a text file). Since you wanted an Excel file, I am attaching the Excel file that is produced by this program.

You will see that the confidence intervals are wide. The confidence intervals for the standard rates would be much wider. The proportion calculated here is probably the best you can do to identify variation in child survival across districts.

- * Program to calculate the proportion of children in the NFHS5 KR file
- * who survived to the date of the survey, in districts in Uttar Pradesh

- * Specify a workspace

```
cd e:\DHS\DHS_data\scratch
```

- * Open the KR file; children born in the past 5 years

```
use "C:\Users\26216\ICF\Analysis - Shared Resources\Data\DHSdata\IAKR7EFL.DTA", clear
```

```
describe v024
```

- * The label for v024 is V024; list it

```
label list V024
```

- * Uttar Pradesh is v024=9

```
keep if v024==9
```

- * How many districts are in Uttar Pradesh?

```
codebook sdist
```

- * 75 districts

```
svyset v001 [pweight=v005], strata(v023) singleunit(centered)  
svy: proportion b5, over(sdist)
```

- * the lines with the output "0" give the proportion of children who died, with a confidence interval

* Save the results and copy into Excel

```
matrix S=r(table)
matrix T=S'
```

* construct a file with the district names and n's, weighted and unweighted

```
gen unwtdn=1
gen wtdn=sweight/1000000
```

```
collapse (sum) *wtdn,by(sdist)
rename sdist district
label values district SDIST
label variable unwtdn "N (unweighted)"
label variable wtdn "N (weighted)"
gen line=_n
sort line
save stub.dta, replace
```

```
clear
svmat T
matrix list T
rename T1 b
rename T2 se
rename T3 t
rename T4 pvalue
rename T5 ll
rename T6 ul
rename T7 df
rename T8 crit
rename T9 eform
gen line=_n
keep if line<=75
sort line
merge line using stub.dta
tab _merge
```

```
rename b P
rename ll L
rename ul U
label variable P "Proportion dead"
label variable L "Low end, 95% c.i."
label variable U "Upper end, 95% c.i."
keep district P L U *wtdn
order district P L U *wtdn
```

```
list, table clean
export excel using child_survival_by_district.xlsx, replace firstrow(var)
```

File Attachments

1) [child_survival_by_district.xlsx](#), downloaded 172 times
